
Subject: Re: the (Moore-Penrose) pseudo-inverse of a matrix - anything like
scipy.linalg's pinv2 in IDL?

Posted by [Heinz Stege](#) on Wed, 03 Apr 2013 13:33:36 GMT

[View Forum Message](#) <> [Reply to Message](#)

On Tue, 2 Apr 2013 22:18:16 -0700 (PDT), JP wrote:

> Is that an equivalent to the scipy pinv2 i am looking for? And if so, I will appreciate if someone
will better algebra skills than me (likely 95% of this community) could suggest how to introduce the
rcond keyword available in pinv2.

>

I am very sure, that I am one of the 5%. So be very careful with the
following code. From the description it looks like the scipy function
is doing something like this:

```
function pinv2,a,rcond=rcond
;
; compile_opt defint32,strictarr,logical_predicate
;
svdc,a,w,u,v ; singular value decomposition
;
n=n_elements(w)
threshold=n_elements(rcond)? max(w)*rcond : 0.
ii=where(w gt threshold,count)
if count lt n then begin
    message,/info,strtrim(n-count,2)+' small singular values.'
    if count le 0 then message,'All singular values are too small.'
end
;
jj=(indgen(n))[ii]*(n+1) ; diagonal elements
matrix=make_array(n,n,type=size(w,/type))
matrix[jj]=1./w[ii]
result=transpose(u)#matrix#v
;
return,result
end
```

If you want to use double precision, take a look at the IDL function
LA_SVD.

Cheers, Heinz
